

## Un driver de LED de puissance

*J'invite le lecteur à consulter le site pour des informations complémentaires.*

Page d'accueil du site Internet :

[page d'accueil](#)

d'autres pdf, sur différents sujets :

[liste des PDF](#)


Les diodes électroluminescentes de puissance nécessitent une électronique de commande adaptée pour faire circuler le courant important nécessaire. La solution industrielle est d'exploiter un convertisseur DC/DC, ce qui, par action sur les temps de conduction / blocage des semi-conducteurs, permet d'obtenir le point de fonctionnement de la LED de puissance tout en minimisant les pertes. Cette solution nécessite un circuit de commande, généralement en circuit intégré dédié. Le schéma montré ici conserve le principe du convertisseur DC/DC, mais en version minimaliste, et, pour réduire les pertes au maximum, sans résistance dans le circuit de puissance.

### 1) Schéma du driver de LED

#### 1.a Principe, fonctionnement qualitatif

On suppose la **source de tension E parfaite**.

Le noyau (M, D, L, C) est un convertisseur DC / DC abaisseur. Le récepteur est la LED de puissance.



L'A. Op., alimenté par une tension simple  $E = 10 \text{ V}$ , est sans contre réaction : sa sortie ne peut prendre que 2 valeurs possibles :  $+V_{\text{SAT}}$  (soit environ  $10 \text{ V}$ ), ou environ  $0 \text{ V}$ . Dans ce cas, le transistor est soit bloqué (car  $|V_{\text{GS}}| = 0 \text{ V}$ ), soit passant (car  $|V_{\text{GS}}| > |V_{\text{TH}}|$ , seuil de conduction), respectivement.

La broche e- est à un potentiel fixe, donné par le pont diviseur 
$$E \frac{R_{p2}}{R_{p1} + R_{p2}} = V_{\text{ref}}$$

Quand le transistor est passant, un courant croît dans L, dont une part circule dans la LED, faisant apparaître une tension à ses bornes. Dès que cette tension dépasse le seuil  $V_{\text{ref}}$ , l'A. Op, monté en comparateur, bascule et sa sortie passe à  $+V_{\text{SAT}}$ , ce qui bloque le transistor. Le courant selfique passe alors dans la diode de roue libre et décroît. Le courant dans la LED décroît, et sa tension repasse sous le seuil de basculement, et l'A. Op. délivre  $0 \text{ V}$  en sortie, ce qui rend à nouveau passant le transistor, et le cycle redémarre.

**Choix de L :** Son rôle est d'assurer une continuité du courant entre le transistor et la diode de roue libre. Intuitivement, une valeur « élevée » est donc conseillée. Pour éviter le phénomène de saturation lié à un circuit magnétique, on oriente le choix vers une self à air, ce qui peut nécessiter un nombre important de spires. Mais, l'encombrement peut être excessif d'une part, et la résistance ohmique également (ce qui engendrerait des pertes par effet Joule). Le compromis ici est  $130 \mu\text{H}$ .


**Choix de C :** Son rôle est d'assurer une tension constante. Une valeur « élevée » est donc conseillée, mais est liée à la durée de la période. Le compromis ici est  $47 \mu\text{F}$ . Remarquons qu'il faut surveiller la résistance série (notée ESR par les constructeurs de condensateurs), qui occasionne des pertes Joule et une fluctuation de tension.

**Choix de la fréquence :** sans circuit de commande dédié, la fréquence n'est pas réglable de façon externe. C'est le choix de L, C, et du seuil de commutation qui fait la durée d'une période. Le pont diviseur  $R_{p1}, R_{p2}$  peut servir de paramètre de réglage pour disposer d'un point de fonctionnement.

**Choix de l'A. Op.** : il faut qu'il travaille sous 10 V, ce qui n'est pas une contrainte forte. Son impédance de sortie doit être faible (pour qu'il charge / décharge la capacité Grille – Source), ce qui est une caractéristique classique de tous les A. Op, même non contre réactionné. Ne pas choisir un slew rate faible, ce qui augmenterait le temps de commutation, et donc les pertes du transistor MOS, lors de la présence simultanée de courant et tension à ses bornes. Il a été choisi le LF411, disponible en librairie.

**1.b. Equations en mode de conduction continue**

Examinons la partie puissance. Le récepteur est ici symbolisé par une résistance. On pose  $\alpha$  le rapport cyclique : durée à l'état passant du transistor / période.




Le potentiel de Drain, en sortie du transistor ne peut prendre que 2 valeurs possibles : E (transistor passant) ou 0 (diode passant) :

Il vient instantanément  $\langle u_D \rangle = \alpha E$ .

En régime périodique,  $\langle u_L \rangle$  vaut 0 V. Par conséquent  $\langle u_C \rangle = \alpha E$ .

D'où le rôle « abaisseur » (de tension) de ce montage.

L'inductance interdit les discontinuités de courant. Sur une période, quand le courant augmente de  $\Delta i$ , il diminue de  $\Delta i$ .



Avec les éléments parfaits, il n'y a aucune perte de puissance. La puissance issue de la source E est transférée intégralement au récepteur.

**Prise en compte des chutes de tensions.**

On peut estimer les semi-conducteurs parfaits lorsqu'ils sont ouverts, mais ils présentent une chute de tension à l'état passant : pour le transistor  $|V_{DS(on)}|$ , et  $|V_{AK}|$  pour la diode de roue libre.

Le calcul de  $\langle u_D \rangle$  devient :  $\frac{1}{T} \left\{ \int_0^{\alpha T} (E - |V_{DS(on)}|) dt - \int_{\alpha T}^T |V_{AK}| dt \right\} = (E - |V_{DS(on)}|) \alpha - |V_{AK}| (1 - \alpha)$

Par conséquent, la tension de sortie est affectée par ces chutes de tension.

**1.c. Composants choisis**

**LED**


On choisit dans la série XP-L2 LED de chez CREELED, la diode électroluminescente XPLBWT.

Le constructeur indique qu'en fonctionnement nominal (85 °C), le courant est de 1,05 A sous une tension directe de 2,79 V :

<https://downloads.cree-led.com/files/ds/x/XLamp-XPL2.pdf>

page 2/34. Il est indiqué également que le courant max est de 3 A.


Le modèle Spice est donné par le fabricant sous : <https://www.cree-led.com/?s=spice>



*LED XPLBWT*

On reporte ci-contre, la caractéristique  $I = f(V)$ , à 85 °C donnée en page 12/34.

On remarque que la puissance maximale peut être de 9 W, si on maintient la température de la puce à 85 °C.




*Caractéristique I = f (V) de la LED de puissance*

### Le transistor MOS

Pour avoir une résistance  $R_{DS(on)}$  de faible valeur, il faut chercher dans les transistors dimensionnés pour les forts courants. L'IRF6540 est donné pour  $I_D = 19$  A continu (sous température de boîtier maintenue à 25 °C), ce qui est surdimensionné pour notre application.

D'après la data sheet de Vishay, en page 2, polarisé à  $|V_{GS}| = 10$  V, il présente une valeur max de  $R_{DS(on)} = 0,2 \Omega$ .



$R_{DS(on)}$  est identifiable par le début de la caractéristique  $|I_D| = f(|V_{DS}|)$  à  $|V_{GS}|$  constant.

Surlignée en rouge, on peut extraire une résistance de

$$\frac{|\Delta V_{DS}|}{|\Delta I_D|} = \frac{10-1}{70-9} = 0,147 \Omega \text{ et ce pour } |V_{GS}| = 10 \text{ V.}$$

Cette valeur est valable pour  $|I_D|$  de 1 à 3 A.





Figure 14. Forward voltage drop versus forward current (typical values) (1N5819)

### Diode de roue libre

On choisit une diode Schottky 1N5819, fabriqué par ST. Une Schottky a l'avantage de présenter une tension de seuil assez faible, ce qui réduit sa puissance dissipée. De plus, c'est à commutation rapide.

La documentation de ce composant montre sa caractéristique statique, présentée ici sur une échelle log log.

Par exemple, on lit ci-contre que parcourue par 1 A, sa tension directe n'est que 0,45 V, ce qui donne une puissance dissipée de 0,45 W.



### Résistances :

On fixe  $R_{p1} = 6,8$  k et  $R_{p2}$  paramétrée, ce qui permet de faire varier la tension  $V_{ref}$ .

## 2) Simulations fichier driver\_LED.cir

- Par un choix du pont diviseur  $R_{p1}$ ,  $R_{p2}$ , on peut faire varier le temps de conduction du transistor, et par là même, passer d'un mode de courant continu à courant discontinu dans l'inductance L.

- Il n'est pas réaliste d'imaginer une température de jonction de la LED (et du transistor MOS) de 25 °C, valeur par défaut du simulateur. Aussi, on fixe arbitrairement une température de tous les composants à 85 °C.

- Le schéma complet comporte 5 noeuds, numérotés de 1 à 5 pour Pspice.

- On simule sur 1 ms, mais pour ne pas avoir la contribution des conditions initiales, on ne montre que l'intervalle [ 0,9 ms ; 1 ms ] représentant le régime périodique.





Schéma simulé

## 2.a. conductions discontinue et continue

### Conductions discontinue obtenue par Rp2 = 2,3 kΩ



*Courant dans l'inductance et courant dans la LED en conduction discontinue*

### Interprétation

Le courant selfique montre des petites variations de quelques mA en dessous de 0. Puis, lors de la mise en conduction du transistor (capturé ici vers 940 μs), croit de façon linéaire jusqu'à 737 mA durant 13,3 μs. Puis décroît vers 0 A durant 35,7 μs.

X Values	939.262u	952.619u
I(L)	-1.4749m	736.458m

Le cycle dure environ 64,4 μs, ce qui laisse environ 15,9 μs de temps mort, où le courant est nul, d'où le nom de conduction discontinue. La fréquence est 15,55 kHz.

Un run sur une longue durée permet d'obtenir sa valeur moyenne : 270,1 mA.

La led reste alimentée grâce à la présence de C. Son courant fluctue par une loi difficile à formaliser, car la caractéristique courant tension est non linéaire. Son courant fluctue entre 97,5 mA et 470,5 mA.


X Values	941.172u	964.964u
I(DLED)	97.515m	470.565m

On peut déterminer, après simulation, sa valeur moyenne : I(DLED) = 272 mA

### Conductions continue obtenue par Rp2 = 2,6 kΩ

C'est le **fonctionnement nominal**. Observons les courants et la tension aux bornes de la LED :

### Courants



*Courant dans l'inductance et courant dans la LED en conduction continue*

**Interprétation :**

Le courant selfique part de 1,015 A, (valeur prise ici à 939,46 μs) et croit de façon linéaire jusqu'à 1,89A durant 16,66 μs. Puis décroît vers 1,01 A durant 36,9 μs. Ce courant circule en permanence, d'où le nom de conduction continue

Le cycle dure 53,48 μs soit une fréquence de 18,7 kHz. Valeur moyenne de I(L) ≈ 1,447 A.

La LED reste alimentée grâce à la présence de C.


Son courant fluctue de 1,1365A à 1,727 A.

On peut mesurer sa valeur moyenne : I(DLED) ≈ 1,448 A.

X Values	939.468u	956.129u
I(L)	1.0148	1.8874

X Values	942.050u	962.960u
I(DLED)	1.1365	1.7270

**Tension**




*Tension aux bornes de la LED*

La tension V(1), aux bornes de C, donc la tension LED, fluctue entre 2,74 V et 2,82 V. Sa valeur moyenne est 2,78 V.

X Values	910.105u	943.158u
V(1)	2.8206	2.7403

La puissance dissipée par la LED est donc 2,78 V x 1,448 A ≈ 4 W.

**2.b. Fonctionnement complet**



*Courants diode, selfique, Source  
Potentiels Drain, Grille (ou sortie de l'A. Op.)*

## Interprétation

### Courants :

On retrouve la notion de conduction continue : quand le transistor est passant, le courant Source passe par l'inductance, ce qui explique  $I_S(M)$  superposé à  $I(L)$ . C'est la phase montante de 1 A à 1,85 A environ. Puis, quand le transistor est bloqué, le courant dans l'inductance passe par la diode de roue libre, ce qui explique  $I(L)$  superposé à  $I(D)$  (phase descendante).

### Tensions :

Quand le transistor est passant, le potentiel de Drain est environ 9,77 V. Le rapport cyclique est de 0,307. Plus finement, en plaçant un curseur à 950  $\mu$ s, on lit  $V(4) = 9,749$  V, soit  $|V_{DSON}| = 0,251$  V, pour un courant  $I_S(M)$  de 1,579 A. Cela donne une résistance  $R_{DSON} = 0,251 / 1,579 = 0,159 \Omega$ , ce qui est en accord avec la valeur extraite en 1.c.

Quand la diode de roue libre est passante, elle présente  $V_{AK}$  allant de 268 mV à 342 mV, soit une valeur moyenne de 0,3 V durant sa durée de conduction.

Vérifions notre équation de tension de sortie :

$(E - |V_{DSON}|) \alpha - |V_{AK}| (1 - \alpha) = (10 - 0,251) \times 0,307 - 0,3 \times 0,693 = 2,78$  V : on retrouve la tension moyenne aux bornes de la LED donnée par la simulation.

### Puissances :

On peut afficher les puissances instantanées, mais, plus intéressant, leur valeur moyenne. Il faut un run très long pour que le calcul de la valeur moyenne converge vers une valeur non liée aux conditions initiales. Il a été choisi 100 ms.

Le bilan des puissances montre que :

- l'alimentation fournit 4,717 W,
- la LED consomme 4,037 W, valeur parfaitement cohérente avec le produit fait précédemment :  $2,78$  V  $\times$   $1,448$  A  $\approx$  4,025 W.

Cela fait un rendement électrique de 85 %.


Les pertes sont de 336,8 mW pour le transistor et de 324,2 mW pour la diode de roue libre. Les autres composants totalisent 19 mW.

X Values	100.000m
- AVG(W(VALIM))	4.7171
AVG(W(DLED))	4.0370
AVG(W(M))	336.829m
AVG(W(D))	324.228m
AVG(W(XTL71))	8.3003m
AVG(W(Rp1))	7.6956m
AVG(W(Rp2))	2.9427m

On voit clairement l'intérêt de cette électronique au lieu d'une simple résistance série : pour avoir le même point de fonctionnement aux bornes de la LED, soit [2,78 V ; 1,448 A], il faudrait une résistance

$$R = \frac{10 - 2,78}{1,448} \approx 5 \Omega \text{ qui dissiperait } 7,24 \text{ W, et l'alimentation devrait fournir } 14,48 \text{ W...}$$

Le rendement serait de  $4 \text{ W} / 14,48 \text{ W} \approx 28 \%$ ...



*une solution simple, mais à mauvais rendement*

### 2.c. Modification du point de fonctionnement par changement de Rp2

On se propose de vérifier la modification de la puissance consommée par la LED, par l'action de Rp2 : le seuil de basculement est modifié par le changement du pont diviseur de tension.

Rp2	2,6 k $\Omega$	2,55 k $\Omega$	2,5 k $\Omega$	2,45 k $\Omega$	2,4 k $\Omega$	2,35 k $\Omega$	2,3 k $\Omega$
<V(1)>	2,7806 V	2,7465 V	2,7099 V	2,6767 V	2,6388 V	2,6067	2,5723 V
<I(DLED)>	1,448 A	1,1845 A	940 mA	722 mA	540 mA	397,3 mA	272,5 mA
P_alim	4,7171 W	3,7695 W	2,9262 W	2,2018 W	1,6133 W	1,1761 W	806,9 mW
P_LED	4,037 W	3,2588 W	2,5537 W	1,9378 W	1,43 W	1,0389 W	702,75 mW
rendement	85,6 %	86,45 %	87,27 %	88 %	88,6 %	88,33 %	87,1 %
fréquence	18,7 kHz	18,36 kHz	17,93 kHz	17,36 kHz	16,59 kHz	16,51 kHz	15,55 kHz

## Interprétation

Ces 7 runs (dont le premier est celui du point de fonctionnement nominal) montrent :

- que la puissance dissipée par la LED est modifiée par la valeur de Rp2,
- que le rendement s'améliore légèrement à plus faible puissance,
- que la fréquence d'oscillation reste sensiblement constante.

Remarque : on quitte le régime de conduction continue pour  $R_{p2} \leq 2,35 \text{ k}\Omega$

## Conclusion

Ce document illustre, sur un exemple concret, l'usage d'un convertisseur DC / DC.

Bien que toutes les valeurs numériques soient parfaitement réalistes, les potentiels et courants peuvent avoir une allure quelque peu différente en pratique. En effet, d'autres considérations doivent être prises en compte, comme la qualité des liaisons entre composants (la ligne de masse doit être dessinée en tenant compte du mixage courants faibles - courants forts), les défauts de l'inductance et du condensateur, etc.

De plus, il est préférable de fabriquer Vref par une référence de tension indépendante de l'alimentation de puissance E. De même, l'A. Op. devrait être alimenté par une source indépendante. En effet, un défaut sur l'alimentation E, comme une inductance série interne de quelques  $\mu\text{H}$ , provoque des oscillations amorties indésirables en V(3) et V(4).

Néanmoins, ce schéma très simple permet d'illustrer et de comprendre un montage connu qu'est le convertisseur DC / DC abaisseur.

### fichier driver\_led.cir

```
Commande de LED par découpage
* Fichier driver_led.cir

.lib eval.lib
.model IRF9540 PMOS(Level=3 Gamma=0 Delta=0 Eta=0 Theta=0 Kappa=0.2 Vmax=0 Xj=0
+ Tox=100n Uo=300 Phi=.6 Rs=64.15m Kp=10.18u W=1.5 L=2u
+ Vto=-3.646 Rd=62.45m Rds=444.4K Cbd=2.029n Pb=.8 Mj=.5 Fc=.5
+ Cgso=1.033n Cgdo=469.4p Rg=.3371 Is=54.08E-18 N=2 Tt=140n)

.model D1N5819 D(Is=2.835u Rs=47.12m Ikf=.3227 N=1 Xti=0 Eg=1.11 Cjo=302.5p
+ M=.7206 Vj=.75 Fc=.5 Isr=16.98u Nr=2)

.MODEL XPLBWT D (IS=4.8985E-27 N=1.721130745045 RS=0.0980008399566
+ EG=2.76169265033408 XTI=11.3767528675097)

.temp=85
VALIM 5 0 DC 10
XTL71 1 2 5 0 3 LF411; e+ e- VPLUS VMOINS Sortie
Rp1 5 2 6.8k
Rp2 2 0 2.6k ; 2.3k
M 4 3 5 5 IRF9540 ; DGSB
L 4 1 130uH
C 1 0 47uF
D 0 4 D1N5819
DLED 1 0 XPLBWT
.IC V(1)=3
.TRAN 0.1u 1m 0 0.1u
.PROBE
.END
```

*Netliste, prête à simuler*

articles 1 à 43 : sur le livre

**Tableau récapitulatif des articles PDF disponibles sur ce site**

n°	titre	lien présentation	lien direct article
	Guide d'installation et d'emploi simplifié	<a href="#">présentation</a>	<a href="#">document PDF</a>
44	Exemples basiques et des exercices...	<a href="#">présentation</a>	<a href="#">document PDF</a>
45	Un exemple de circuit passif	<a href="#">présentation</a>	<a href="#">document PDF</a>
46	Un oscillateur Colpitts	<a href="#">présentation</a>	<a href="#">document PDF</a>
47	Compensation en fréquence des amplificateurs opérationnels	<a href="#">présentation</a>	<a href="#">document PDF</a>
48	Un amplificateur à transistors bipolaires	<a href="#">présentation</a>	<a href="#">document PDF</a>
49	Une bascule D Flip Flop CMOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
50	Une porte XOR à transistors MOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
51	Un VCO à 12 transistors MOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
52	Une PLL à moins de 20 transistors MOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
53	Un oscillateur à résistance négative	<a href="#">présentation</a>	<a href="#">document PDF</a>
54	Une charge électronique	<a href="#">présentation</a>	<a href="#">document PDF</a>
55	Un amplificateur en classe C	<a href="#">présentation</a>	<a href="#">document PDF</a>
56	Le monostable 74 123	<a href="#">présentation</a>	<a href="#">document PDF</a>
57	Un amplificateur en classe D	<a href="#">présentation</a>	<a href="#">document PDF</a>
58	Le transformateur en linéaire	<a href="#">présentation</a>	<a href="#">document PDF</a>
59	La loi d'ohm thermique	<a href="#">présentation</a>	<a href="#">document PDF</a>
60	Le transformateur en non linéaire	<a href="#">présentation</a>	<a href="#">document PDF</a>
61	Robustesse d'un oscillateur en anneau	<a href="#">présentation</a>	<a href="#">document PDF</a>
62	Une alimentation stabilisée	<a href="#">présentation</a>	<a href="#">document PDF</a>
63	Modélisation d'un haut-parleur	<a href="#">présentation</a>	<a href="#">document PDF</a>
64	Un synthétiseur de fréquence	<a href="#">présentation</a>	<a href="#">document PDF</a>
65	Un ampli audio de Sparkfun	<a href="#">présentation</a>	<a href="#">document PDF</a>
66	Simulation logique et analogique	<a href="#">présentation</a>	<a href="#">document PDF</a>
67	Un oscillateur à relaxation	<a href="#">présentation</a>	<a href="#">document PDF</a>
68	Lecteur de TAG RFID 125 kHz	<a href="#">présentation</a>	<a href="#">document PDF</a>
69	Diagramme de l'œil avec Pspice	<a href="#">présentation</a>	<a href="#">document PDF</a>
70	Un amplificateur hautes fréquences	<a href="#">présentation</a>	<a href="#">document PDF</a>
71	Une bizarrerie enfin expliquée...	<a href="#">présentation</a>	<a href="#">document PDF</a>
72	Comprendre le paramétrage de la FFT	<a href="#">présentation</a>	<a href="#">document PDF</a>
73	La relation de Bennett	<a href="#">présentation</a>	<a href="#">document PDF</a>
74	Simuler un circuit à plus de 20 transistors avec PSpice Eval	<a href="#">présentation</a>	<a href="#">document PDF</a>
75	Une horloge biphasé sans recouvrement	<a href="#">présentation</a>	<a href="#">document PDF</a>
76	Quelques simulations sur la diode	<a href="#">présentation</a>	<a href="#">document PDF</a>
77	Un ampli classe A, avec transformateur de sortie	<a href="#">présentation</a>	<a href="#">document PDF</a>
78	Des stimuli pour PSpice	<a href="#">présentation</a>	<a href="#">document PDF</a>
79	Simuler le TL431 : zener ajustable	<a href="#">présentation</a>	<a href="#">document PDF</a>
80	Un ADC flash	<a href="#">présentation</a>	<a href="#">document PDF</a>
81	Une chaîne d'acquisition : S&H, ADC, DAC	<a href="#">présentation</a>	<a href="#">document PDF</a>
82	Un amplificateur 50 MHz	<a href="#">présentation</a>	<a href="#">document PDF</a>
83	Un dérivateur non inverseur	<a href="#">présentation</a>	<a href="#">document PDF</a>
84	Un amplificateur bipolaire avec push pull CMOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
85	Rôle des répéteurs logiques dans un circuit intégré	<a href="#">présentation</a>	<a href="#">document PDF</a>
86	Un driver logique CMOS pour charge 50 ohms	<a href="#">présentation</a>	<a href="#">document PDF</a>
87	Des triggers de Schmitt et des applications	<a href="#">présentation</a>	<a href="#">document PDF</a>
88	Un filtre gaussien analogique	<a href="#">présentation</a>	<a href="#">document PDF</a>
89	Un générateur de bruit rose	<a href="#">présentation</a>	<a href="#">document PDF</a>

90	Un anémomètre à fil chaud : simulation comportementale	<a href="#">présentation</a>	<a href="#">document PDF</a>
91	Un oscillateur à pont de Wien stabilisé par CTN	<a href="#">présentation</a>	<a href="#">document PDF</a>
92	L'emballement thermique d'une diode	<a href="#">présentation</a>	<a href="#">document PDF</a>
93	Les puissances dans un amplificateur	<a href="#">présentation</a>	<a href="#">document PDF</a>
94	Asservissement de puissance dans une résistance	<a href="#">présentation</a>	<a href="#">document PDF</a>
95	Asservissement de la puissance émise par une antenne radio	<a href="#">présentation</a>	<a href="#">document PDF</a>
96	Un driver de LED de puissance	<a href="#">présentation</a>	<a href="#">document PDF</a>
97	Exploiter Pspice pour simuler des filtres numériques	<a href="#">présentation</a>	<a href="#">document PDF</a>
98	Un filtre en cosinus surélevé avec Pspice	<a href="#">présentation</a>	<a href="#">document PDF</a>
99	Effet de la température sur un amplificateur en classe A	<a href="#">présentation</a>	<a href="#">document PDF</a>
100	Un amplificateur à transistors JFET et bipolaires	<a href="#">présentation</a>	<a href="#">document PDF</a>
<b>Supplément, hors article :</b>			
<b>mon cours « Electronique pour les communications numériques », <a href="#">polycopié couleur 201 pages en pdf</a></b>			

[retour à l'écran d'accueil de ce site](#)